Tutorial 9: 
Multiplications with Accumulation

New Instructions Introduced

MAC

MACA

New Flags Introduced

OVM

Overview of Tutorial

There are several DSP applications, such as finite impulse filters (FIRs) that require a large number of sum-of-products terms. This is achieved on the TMS320C5000 by carrying out a series of multiplications, sometimes hundreds, and adding the products together. This process is referred to as multiplication with accumulation and its implementation using the TMS320C5000 is described in this tutorial.    

Generating a Sum-of-Products

To implement a finite impulse filter (FIR) we multiply a series of inputs with a series of constants. A typical format is as shown in Equation 9-1:

Equation 9-1. 

Where N = number of coefficients, values h(0), h(1) … are the coefficients, x(n) is the input at time n and x(n-1)… , x(n-N) are delayed input samples. 

Consider the case where we have eight inputs, each of which is to be multiplied by the corresponding entry from a table. We can implement this in C code as shown in Example 9-1: 

Example 9-1.

	
	

	unsigned int input[8];
	// Eight inputs.

	unsigned int table[8]={10,15,22,27,

                       27,22,15,10 };
	// Table of

// constants.

	unsigned int i;
	// Iterations

// counter.

	unsigned long accumulator_A;
	

	unsigned long accumulator_B = 0;
	

	for ( i = 0; i < 8 ; i++)
	

	{
	

	accumulator_A = input[I] * table[i];
	// Multiply one 

// element.

	accumulator_B += accumulator_A;
	// Accumulate in

// accumulator B.

	}
	// Final result in

// accumulator B.


We have a series of eight input samples that we multiply by eight constants. The usual way to do this in C code is to use two arrays, one for the inputs and one for the constants. We multiply each input by the corresponding constant and then add the product to the accumulated total.  

Accumulation Using Direct Addressing

Let us take an example of multiplication with accumulation using TMS320C5000 assembly language. We wish to multiply a series of 16 data words beginning at data memory address 80h by another series of data words beginning at data memory address 90h. The result of each multiplication is added to the previous and the final result is to be stored in accumulator B.

Using direct addressing we may write part of the code sequence as shown in Example 9-2:

Example 9-2.

	
	

	LD #1, DP
	; Page 1. Gain access to data memory

; addresses 80h to FFh.

	
	

	LD 0h, T
	; Load the T register with the contents of

; the data memory address 80h + 0h = 80h.

	MPY 10h, B
	; Multiply the T register by the contents of

; the data memory address 80h + 10h = 90h.

; The product is placed in accumulator B.

	
	

	LD 1h, T
	; Load the T register by the contents of

; the data memory address 80h + 1h = 81h.

	MPY 11, A
	; Multiply the T register by the contents of

; the data memory address 80h + 11h = 91h.

; Product is placed in accumulator A.

	ADD A, B
	; Add accumulator A to accumulator B and 

; put result in accumulator B.

	
	

	
	; Repeat LD / MPY / ADD A instructions as

; required. 


There is a major disadvantage in using direct addressing when carrying out a large number of multiplications with accumulation. Each data memory address must be written out in full and this means a large number of lines of code are required. A better choice is offered by indirect addressing.    

Using Indirect Addressing

Let us now repeat the same operation we carried out in Example 9-2, but this time using indirect addressing. For this we will use auxiliary register AR1 as a pointer to the data memory block at 80h and auxiliary register AR2 as a pointer to the data memory block at 90h. The first task is to the clear registers and initialize the  pointers:  

Example 9-3.

	
	

	STM #80h, AR1
	; Store address of start of data memory

; block in AR1. AR1 = 80h.

	STM #90h, AR2
	; Store address of start of data memory

; block in AR2. AR2 = 90h.

	LD #0, A
	; Clear accumulator A so accumulations

; start at zero.


Initialization has been done. Now we start a sequence of multiplications with accumulation: 

Example 9-4.

	
	

	LD *AR1+, T
	; Load the T register with the contents

; of the data memory address pointed to

; by auxiliary register AR1. AR1 = 80h.

; Increment auxiliary register AR1 to

; point to the next data memory address

; AR1 = 81h.   

	MPY *AR2+, A
	; Multiply the T register by the contents

; of the data memory address pointed to

; by AR2. AR2 = 90h. Put the product in

; accumulator A. Increment AR2 to point

; to the next data memory address. 

; AR2 = 91h.  

	ADD A, B
	; Add the product of this multiplication

; to the total sum-of-products in

; accumulator B.  


In Example 9-4, we have used a variation of the instruction LD to load the contents of a data memory address into the T register. 

The block of instructions in Example 9-4 would be executed 16 times to take us through successive data memory addresses. 

When the final multiplication and accumulation has been done, AR1 will contain the value 90h and AR2 will contain the value A0h. Hence AR1 and AR2 will be pointing to the data memory addresses one past their relevant data blocks. This does not normally pose a problem, provided that AR1 and AR2 are re-initialized before they are next used.

Using a Loop Counter for Multiple Accumulations

The method of writing each multiplication as inline code will give the fastest execution time, but in terms of code efficiency, a better method is to use a loop counter. The instruction BANZ (branch if auxiliary register non-zero) provides a convenient means of executing a control loop. We shall combine together the operations shown in Examples 9-3 and 9-4 and use auxiliary register AR3 as a loop counter, as illustrated in Example 9-5:

Example 9-5.

	
	
	

	
	LD #0, B
	; Clear accumulator B

	
	STM #80h, AR1
	; Store address of start of first

; data block in auxiliary register

; AR1. AR1 contains 80h.

	
	STM #90h, AR2
	; Store address of start of first

; data block in auxiliary register

; AR2. AR2 contains 90h.

	
	STM #15, AR3
	; AR3 is our loop counter. 16

; iterations (16 - 1 = 15

; loops).

	lp1:
	LD *AR1+, T
	; Load the T register with the

; value from the first data block. ; Point to next address in first

; data block. 

	
	MPY *AR2+, A
	; Multiply the T register by the

; value from second data block.

; Put product into accumulator A.

; Point to next address in second

; data block.

	
	ADD A, B
	; Accumulate product into

; accumulator B. 

	
	BANZ lp1, *AR3- 
	; Test if auxiliary register AR3

; contains zero. If so, the loop

; is complete. Otherwise branch to

; the label lp1. In both cases,

; decrement the loop counter AR3.


Here lp1 is a label indicating the beginning of the loop and is used by the instruction BANZ (branch auxiliary register non-zero). Note that we have used three auxiliary registers: AR1 as a pointer to the first block, AR2 as a pointer to second block and AR3 as the loop counter.  

Combining  Instructions 

Consider the following two instructions used as part of a multiplication with accumulation sequence: 

Example 9-6.

	
	

	MPY *AR2, A
	; Multiply the T register by the value

; pointed to by AR2. Put the product into

; accumulator A.

	ADD A, B
	; Add product in accumulator A to total in 

; accumulator B. 


In the case of Example 9-6 we have put the product of the multiplication into accumulator A then added it to accumulator B. These two instructions can be combined into the single instruction MAC (multiply and accumulate), as shown in Example 9-7:

Example 9-7.

	
	

	MAC *AR2, B
	; Multiply the T register by contents of the

; data memory address pointed to by

; AR2. Add the product to the value in

; accumulator B and store that sum in

; accumulator B.


The instruction MAC (multiply with accumulate) takes two operands. The first operand is the address in data memory of the source. The second operand is the destination accumulator where we are carrying out the accumulation. Not only does the method used in Example 9-7 save an instruction, but it has the additional benefit of only using one accumulator.

Preventing Overflow 

One of the potential problems when carrying out a large number of multiplications with  accumulation is that an overflow can occur in the accumulator. This happens when the result of the addition exceeds the greatest value that can be stored in that accumulator. 

For example, say accumulator A contains the largest signed 32-bit value of 7FFFFFFFh (+2147483647) and then we add one. Accumulator A will now contain the value 80000000h (-2147483648). This represents a negative number. The largest signed 32-bit number has been exceeded and the counter has wrapped around to the most negative number. This is known as overflow. 

Similarly, if accumulator B contains the smallest 32-bit value of 8000000h 

(-2147483647) and we subtract one, then accumulator B wraps around to contain 7FFFFFFFh (+247483647). This represents a positive number. An overflow has occurred in the opposite direction, and is sometimes referred to as underflow.   

When an overflow of an accumulator occurs, the contents become unreliable; the result cannot be trusted to be the correct value or even to have the right sign.    

There is a defensive mechanism available on the TMS320C5000 of being able to put a limit on the value that is stored in an accumulator. This is known as saturation. Saturation means that the largest positive value that an accumulator can contain is 7FFFFFFFh and the most negative value that an accumulator can contain is 80000000h. To avoid overflow we make use of the OVM bit in status register ST0, as shown in Example 9-8:

Example 9-8.

	
	

	RSBX OVM
	; Turn off overflow mode (OVM). Accumulators

; are allowed to wrap around. 

	SSBX OVM
	; Turn on overflow mode (OVM). Prevent the

; accumulators overflowing by saturating

; them at either 7FFFFFFFh or 80000000h.


Let us now look at Example 9-9 which shows what happens when overflow mode is turned off:

Example 9-9.

	
	

	SSBX SXM
	; Turn on sign-extension mode.

	RSBX OVM
	; Turn off overflow mode. Allow

; overflows.

	LD #7FFFh, 16, A
	; Load high word of accumulator A with

; 7FFFh. Accumulator A contains 

; 7FFF0000h.

	OR #0FFFF, A
	; Logical OR with FFFFh. Accumulator A

; now contains 7FFFFFFFh (+247483647). 

	ADD #1, A
	; Add 1 to accumulator A. Accumulator A

; now contains 80000000h (-2147483648).

; There has been an overflow. 


Now let us see what happens when overflow mode (OVM) is turned on, as is the case in Example 9-10:

Example 9-10.

	
	

	SSBX SXM
	; Turn on sign-extension mode.

	SSBX OVM
	; Turn on overflow mode. Saturate

; accumulator rather than allow

; overflow. 

	LD #7FFFh, 16, A
	; Load high word of accumulator A with

; 7FFFh. Accumulator A contains

; 7FFF0000h

	OR #0FFFFh, A
	; Logical OR accumulator with FFFFh.

; Accumulator A now contains +247483647

; (7FFFFFFFh). 

	ADD #1, A
	; Add 1 to accumulator A. Accumulator A

; still contains 7FFFFFFFh. Saturation

; has occurred. 


When overflow mode is turned on using the instruction SSBX OVM, overflow is prevented. This means that the value in the accumulator being used is limited to a maximum of 7FFFFFFFh or a minimum of 80000000h. 

Note that when saturation occurs, the accumulator does not contain the correct value, but the percentage error will be small, and most important, the sign will be correct. In a later tutorial we shall see how to test for overflow, assuming overflow mode is turned off, and thus take appropriate action.  

From this point on in the tutorial, we will recognize that overflow could occur on accumulation and therefore set OVM (overflow mode). 

Final Code Reduction
So far we have seen how to use the instruction MAC (multiply accumulate) to multiply the T register by a value stored in data memory. We can also use the instruction MAC to multiply together two variables stored in data memory and accumulate the result in one of the accumulators. Example 8-11 shows how to multiply together the contents of two data memory addresses and then accumulate the result:    

Example 9-11.

	
	

	LD *AR1, T
	; Load the T register with the contents

; of the data memory address pointed

; to by auxiliary register AR1.   

	MPY *AR2, A
	; Multiply the T register by the contents

; of the data memory address pointed to

; by AR2. Put the product in accumulator

; A.  

	ADD A, B
	; Add the product of this multiplication

; to the total sum-of-products in

; accumulator B.  


The three instructions in Example 9-11 can be condensed into a single instruction:

Example 9-12.

	
	

	MAC *AR1, *AR2, B
	; Load the T register with the

; contents of the data memory address

; pointed to by auxiliary register

; AR1. Multiply the T register by the

; contents of the data memory address

; pointed to by AR2. Add this product

; to the sum-of-products in

; accumulator B.    


The fact that we have been able to combine the instructions LD, MPY and ADD into a single instruction is significant. In order to repeat a series of three instructions in a loop means we must make use of an instruction such as BANZ (branch on auxiliary register non-zero to control the loop. However, to repeat a single instruction, we can use the instruction RPTZ (clear accumulator then repeat next instruction).

Let us revisit the earlier problem of multiplying a series of 16 data words beginning at data memory address 80h with another series of data words beginning at data memory  address 90h and put the sum-of-products in accumulator B. We can now do this in a more efficient way, as shown in Example 9-13:

Example 9-13.

	
	

	SSBX OVM
	; Prevent overflows by allowing

; saturation.

	STM #80h, AR1
	; Store the address of the start of

; the first data block in auxiliary

; register AR1. AR1 = 80h.

	STM #90h, AR2
	; Store the address of the start of

; the second data block in auxiliary

; register AR2. AR2 = 90h.

	RPTZ B, #(16-1)
	; Clear accumulator B prior to

; accumulation and execute the

; following instruction 16 times.

	MAC *AR1+, *AR2+, B
	; Load the T register with the

; contents of the data memory 

; address pointed to by auxiliary

; register AR1. Multiply the T

; register by the contents of the

; data memory address pointed to by

; AR2. Add this product to the

; sum-of-products in accumulator B.

; Increment AR1 and AR2.    


The operand used with the instruction RPTZ is the number of repeats, and hence is one less than the total number of operations.

We have used the operands *AR1+ and *AR2+ so that the two auxiliary registers  being used as pointers are incremented each time the instruction MAC is executed.   

This final code reduction illustrates the power of the TMS320C5000 instruction set. We have been able to implement a complex function using very few instructions. 

Multiplication with Accumulation using an Accumulator

There is a second variation on the multiply and accumulate instruction, this time using accumulator A instead of the T register:

Example 9-14.

	
	

	STM #100h, AR4
	; Store the address of the data

; memory location in auxiliary

; register AR4. AR4 = 100h.

	MACA *AR4+, B
	; Multiply the contents of the data

; memory address pointed to by AR4

; by the high word of accumulator A

; then accumulate in accumulator B.


The instruction MACA (multiply by accumulator A and accumulate) takes two operands. The first operand is the source data memory address and the second operand is the accumulator used as the destination.   

When using the instruction MACA, accumulator A is taken to be a 17-bit unsigned value. Note that the high word of accumulator A is used, rather than the low word. When accumulator A is loaded with a constant, a shift of 16 to the left must be used, otherwise the product of the multiplication will be incorrect. The correct usage is shown in Example 9-15:

Example 9-15.

	
	

	LD #0, DP
	; Page 0. Gain access to data memory

; addresses 60h to 7Fh.

	RSBX SXM
	; Turn off sign-extension mode.

	LD #100h, 16, A
	; Load accumulator A with value 100h

; shifted 16 places to the left.

; Accumulator A now contains 01000000h.

	MACA 70h, B
	; Multiply the contents of the data

; memory address 70h by the high

; word of accumulator A and accumulate

; in accumulator B.


Should the shift be omitted when loading the accumulator, then the product of the multiplication contains the wrong value. 

If sign-extension mode is turned on, then the high word of accumulator A can contain FFFFh, rather than the immediate value. 

Example 9-16.

	
	

	LD #0, DP
	; Access to data memory addresses

; 60h to 7Fh.

	RSBX SXM
	; Turn off sign-extension mode.

	LD #8000, A
	; Incorrect. Shift omitted.

; Load accumulator A with 8000h.

; Accumulator A now contains FFFF8000h.

	MACA 70h, B
	; Multiply the contents of the data

; memory address 70h by the high

; word of accumulator A (FFFFh). The

; wrong value will be accumulated in

; accumulator B. 


If sign-extension mode is turned off, the high word of the accumulator will be loaded with 0000h and the product accumulated will always be zero.

Example 9-17.

	
	

	LD #0, DP
	; Access to data memory addresses

; 60h to 7Fh.

	RSBX SXM
	; Turn off sign-extension mode.

	LD #8000h, A
	; Incorrect. Shift omitted. 

; Load accumulator A with 8000h. 

; Accumulator A now contains 00008000h.

	MACA 70h, B
	; Multiply the contents of the data

; memory address 70h by the high

; word of accumulator A (0000h). Zero

; will be accumulated in accumulator B. 


Upgrading from the TMS320C2000 to the TMS320C5000

As far as multiplications with accumulations are concerned, the architecture of the TMS320C5000 differs from that of the TMS320C2000. There is no P register so that the instructions LTA (load T register and accumulate previous product), APAC (add product register to accumulator), and SPM (set product register shift mode) are not required.

Table 9-1. Comparison of Instructions

	Description
	TMS320C2000

Instruction
	TMS320C5000

 Instruction

	Load T register and accumulate previous product
	LTA
	Not used

	Add product register to accumulator
	APAC
	Not used

	Set product register shift mode
	SPM
	Not used

	Multiply and accumulate
	MAC
	MAC *ARx, Acc

	Multiply accumulator and accumulate
	Not available
	MACA *ARx, Acc


ARx = One of the auxiliary registers AR0 to AR7.

Acc = Accumulator A or accumulator B. 

On the TMS320C2000, one way to avoid overflow is to shift the product six places to the right before accumulating. This effectively divides the product by 26 before accumulating it and so causes a loss of resolution.

The behavior during overflow has been improved on the TMS320C5000 by the use of  40-bit accumulators. This means that it is possible to carry out multiplication with accumulation on variables greater than 32 bits. 

Questions

	1.
	What is meant by the term accumulation?

	2.
	When carrying out a series of multiplications with accumulations, why might we use indirect addressing in preference to direct addressing?

	3.
	What is meant by the term overflow?

	4.
	Why can overflow be a problem when carrying out accumulation?

	5.
	How do we prevent overflow when carrying out accumulation?

	6.
	What are the advantages of the instruction MAC over the instructions MPY and ADD?

	7.
	What is meant by the term saturation?

	8.
	How do we turn on overflow mode?

	9.
	How do we turn off overflow mode?

	10.
	Combine the following two instructions into a single instruction:

MPY *AR3+, A

ADD A,B

	11.
	Combine the following three instructions into a single instruction:

LD *AR3, T
MPY *AR4, A

ADD A, B

	12.
	Why will the following code always produce a product of zero in accumulator B?

RSBX SXM 

LD #200h, A

MAC 80h, B

